
INTRODUCTION
The option to adapt applications without the need of modification is one of the main benefits offered by FPM.
The possibilities of Web Dynpro Customizing in combination with the Enhancement Framework already allow
far more than the classic Dynpro in this respect.

Nevertheless, these two options still have some limitations: there is only one customizing per client, and
Enhancements are system-wide active. This allows for adapting applications to the needs of a customer, but
not to the needs of different contexts within one customer system.

Example

With Customizing and Enhancing, it’s quite simple to adapt the above screen. The customer can rearrange,
hide fields, or even add additional fields. However, he must decide how this screen should look like in any
situation…and he is not able to adapt it to some runtime parameter.

The need to add additional information to the screen for particular users might arise (or, for example, to avoid
overstretching managers, the detail assignment block and some of the fields could be removed). So far, to
achieve this, there was only the option to copy the whole set of configurations, adapt it and let the users start
the adapted configuration.

Another problem is the need for data-dependent adaptations. In the above screen, the “Street/House
Number” fields are placed in the right order for German addresses. In France, House Number and Street
Name are exchanged.  Therefore, an adaptation for French addresses would simply exchange those two
fields. However, so far this can only be realized in a very cumbersome way: the Form configuration has to be
replaced at runtime via AppCC coding.

Context-Based Adaptations (CBA) provide you with a far better and more comfortable approach to
implement these use-cases.

BASIC CONCEPTS
Before going into a detailed step-by-step description, some basic terms need explaining:

Adaptation Schema
The adaptation schema is simply a list of characteristics (or dimensions) which can be used for adaptations.
In the example above, there is the need for two characteristics: role and country. The adaptation schema is
created using SM30 views FPM_V_ADAPT_SCHM and FPM_V_ADAPT_DIM.

Adaptation Dimension
The adaptation dimension represents an individual characteristic within an adaptation schema. It is
maintained in view FPM_V_ADAPT_DIM. A dimension is defined by a name for identification purposes, an
index, and a data element. The index is used to determine the dominating adaptation dimension in possible
collision cases. The data element is only used in the design-time, allowing you to provide field and F4 helps
while configuring context-based adaptations.

Adaptation Context
The adaptation context is a set of characteristics values for a given adaptation schema. So, if the adaptation
schema TEST consists of the two dimensions ROLE and COUNTRY, a sample adaptation context would be
ROLE = MANAGER and COUNTRY = FRANCE.

The term “context” in this document is completely independent of the Web Dynpro (WD) term
“context”. To avoid misunderstandings, the term “adaptation context” is used.

Inheritance of Component Configurations
Context-Based Adaptations are using the inheritance concept of component configurations. Each adapted
configuration is represented by a derived configuration. A derived configuration only contains the delta to its
base configuration.

STEP-BY-STEP EXAMPLE
In the first part of this example we will extend the existing demo WD application S_EPM_FPM_PO by adding
an Attachment UIBB to the overview page. This additional UIBB should only be available for Managers. As
this can be done without any modification of the existing application, there is no need to create a copy first.
In the second part of this example we will exchange street and house number, appropriate for French
addresses.

Adding an Attachment UIBB for Managers

1. Create the Adaptation Schema
As we want to adapt the application with respect to role (only Managers should see the attachment UIBB)
and country (French addresses need a different field order), we need to create an adaptation schema having
the dimensions Role and Country.  As such an adaptation schema already exists (TEST_FPM), you can use
this one and skip this step.

To do this, launch SM34, enter FPM_VC_ADAPT_SCHEMA as View Cluster, and create a new adaptation
schema. Then add the two required dimensions.

An adaptation schema should normally be valid throughout a whole application area; there is no need to
create an individual schema for every application.

2. Create an Adaptable Application
To run S_EPM_FPM_PO as an adaptable configuration, a new WD application must be created. This can be
done quite easily by running the CBA Enabler wizard as follows:

Launch transaction SE80 to open the FPM Workbench. Click the link Enable FPM Applications for Context-
Based Adaptations to start the CBA Enablement Wizard. The following screen appears:

Enter the name of the original application (S_EPM_FPM_PO), the Adaptation Schema (TEST_FPM), the
name of the new CBA-enabled application, and a name for its application configuration (e.g.
Z_CBA_EPM_FPM_PO for both). Press Next and, on the next screen, Save.

This will bring you to the confirmation screen of the CBA Enabler (see the following screen).

There are two links. The first link will take you to the configuration editor, and the second will launch the
application. Don’t close this window as you will need these links in the following steps.

Click the second link. This will start your new application. It should look exactly the same as the original
application; we only enabled it for CBA, but we didn’t adapt anything yet.

3. Create an Adaptation
Click the first link on the confirmation screen, this will launch the FPM configuration editor (FLUID).  The
adaptation panel is probably hidden. Unhide it by toggling the corresponding Adaptations toggle button in the
main toolbar. The Adaptations panel appears directly below the main toolbar:

There’s already one entry, the base configuration, which you now can adapt.

To create an adaptation for the Manager, press the Add button. In the Add Adaptation popup that appears,
enter MANAGER as the Role. Leave the Country field empty.

After pressing OK and going through the package popup(s) - if in doubt, choose Local Object - you will get a
second entry in the adaptation panel. Select it and make sure you’re in edit mode.

Now add a new UIBB to the page and save it. The following screenshot shows an example result:

If you ran the application now, it would be unchanged. This is because we have still not set an adaptation
context. However, the application is now prepared to show the additional UIBB for when the adaptation
context is set to role = MANAGER.

This can now easily be achieved by adding the relevant adaptation context as URL parameters. Therefore,
simply add &ROLE=MANAGER to the URL of the application. You should then see that the application has
been adapted, and the attachment UIBB is displayed in the bottom of the overview page.

Before we go on to the second part of this example, let’s just have a quick summary of what we’ve just done.
We started with a SAP-delivered standard application and adapted it. We had to work through a couple of
different steps, but we did not have to code or modify existing objects. All the steps were purely
configurative.

Adapting the Address Layout
In the second part we will try to have some dynamic adjustment of the screen layout based on the data to be
displayed. As the chosen application is not yet prepared to support CBA, this second part will require a small
bit of coding.

As we’re using the same application, we can skip the steps 1-3 from the previous part and directly continue
with preparing the application to support CBA.

1. Extend the Form’s Feeder Class
The data in the original application’s form is provided by feeder class CL_EPM_PO_FORM_FEEDER. To
make the form adaptable regarding country, the feeder class must communicate the country of the data to
the CBA framework.

The original feeder class doesn’t do this because it was created before CBA was available.

To provide the CBA functionality without modifying the standard feeder class, we need a new feeder. As we
don’t want to reimplement all its functionality, we simply subclass it and only redefine the one method.

Therefore, go to SE24 and create a new class that inherits from CL_EPM_PO_FORM_FEEDER. The only
method we must redefine is the IF_FPM_GUIBB_FORM~GET_DATA method, and implement it using the
following code:

method IF_FPM_GUIBB_FORM~GET_DATA.
* Needed to convince feeder that data needs to be updated

if cs_data is initial.
      get_shared_data_instance( cl_epm_po_shared_data=>gc_class_name

)-
>add_all_change_notification( if_epm_po_header=>gc_bo_node_name ).

endif.

* call the GET_DATA method of the standard feeder class
call method super->if_fpm_guibb_form~get_data
exporting

        io_event = io_event
        it_selected_fields = it_selected_fields
        iv_raised_by_own_ui = iv_raised_by_own_ui
        iv_edit_mode = iv_edit_mode

importing
        et_messages = et_messages
        ev_data_changed = ev_data_changed
        ev_field_usage_changed = ev_field_usage_changed
        ev_action_usage_changed = ev_action_usage_changed

changing
        cs_data = cs_data
        ct_field_usage = ct_field_usage
        ct_action_usage = ct_action_usage.

* Check event id to avoid infinite loop
check io_event->mv_event_id ne

                    if_fpm_constants=>gc_event-adapt_context.

data: lo_fpm type ref to if_fpm,
          lo_event type ref to cl_fpm_event.

field-symbols: <fs_country_code> type snwd_country.
* determine the country code from the data delivered by the standard feeder

assign component 'COUNTRY_CODE' of structure cs_data
to <fs_country_code>.

* create the object to set the adaptation context. .
create object lo_event
exporting

        iv_event_id = if_fpm_constants=>gc_event-adapt_context
        iv_adapts_context = abap_true.

* Set the adaptation context via event parameters
    lo_event->mo_event_data->set_value(

exporting iv_key = 'COUNTRY'
                                  iv_value = <fs_country_code> ).

* finally raise the event
    lo_fpm = cl_fpm_factory=>get_instance( ).
    lo_fpm->raise_event( lo_event ).

endmethod.

Let’s walk through the code step-by-step:

1. The first lines are very specific for this example. Don’t think too much about them as they are only
necessary for this special example (the parent feeder class has some special delta mechanism which
only updates the data in specified situations – without this line, the parent feeder class wouldn’t return any
data).

2. We then call the IF_FPM_GUIBB_FORM~GET_DATA of the parent class, making sure that the standard
logic and data retrieval is executed. The logic of the feeder class remains unchanged. All we do is add
some additional lines of code to inform the CBA framework about the changed adaptation context.

3. Then we check if the current event isn’t an adaptation event.  This is because setting the adaptation
context is done by raising another FPM event that then calls this method again. Omitting this check would
cause an infinite loop.

4. Next, we determine the country from the data provided by the standard implementation.

5. Finally, we create an event to change the adaptation context, adding the country information as event
parameter and raising the event.

Please note that this code is far from being optimized. It is kept as simple as possible for the example to
work and to demonstrate how CBA works. A better version will be provided in a later chapter.

2. Replace the Standard Feeder Class
Now we will use a trick to replace the standard feeder class with the one we created. To do this, create an
adaptation of the original form configuration. Call up your new application configuration again and navigate
from there via the OVP component configuration to the Form UIBB configuration. Create a new adaptation
for it. Add adaptation context set ROLE = * and leave the Country value empty. Then replace the feeder
class within the adapted configuration.

ROLE = * means this adaptation should be applied to all adaptation contexts, where no more specific
adaptation exists. As, currently, this is the only adaptation of the form, this adaptation will always be applied.
Therefore, this is a simple trick to replace a feeder class without modifying any standard object.

These two steps will be much easier if you want to prepare your own application for CBA. In that case, you
would simply add the necessary lines of code to the standard feeder application - and that’s it! There’s no
need to create your own feeder class and replace it via an adaptation.

Applications that want to provide the adaptation functionality should add the necessary code to their
standard code. This way, customers are completely enabled for CBA without these more complex steps.

3. Create an Adaptation for French Addresses
 The last step is to create an adaptation of the form especially for the context COUNTRY = FR. So, navigate
again to the Form’s configuration via the application and the OVP component configuration. Create a new
adaptation and set the adaptation context to ROLE = * COUNTRY = FR.

Replace the feeder class and add your own and exchange the street and house number fields.

Run the application. If you select a French supplier, you should now see the adjusted configuration.
Now try to run your application again, but this time add ROLE=MANAGER to the start URL. The result is
probably no surprise: You get the additional Attachment UIBB because you started the application as a
Manager, and for French addresses you get the right order for Street/House Number.

4. Enhance the French Address Adaptation
As already mentioned earlier, there is some room for improvement in this example. There are two issues with
the current solution:

· The feeder class is currently firing an additional event to set the adaptation context at every FPM event.
This means that the number of FPM roundtrips is doubled. And it gets worse. If you assume that it should

be possible to run the feeder class in an application which doesn’t support CBA, then, even in these
applications the number of FPM events is doubled. As this might lead to performance implications, it
should be avoided.

· The adaptation context is set globally. This means it will be applied to all UIBBs on the screen. This is
normally OK. However, there could be instances where another UIBB on the screen uses another
adaptation context.

a. Avoiding Unnecessary FPM Events

This can, of course, be solved very easily.  To avoid that this CBA-specific code is executed in applications
that don’t support CBA, you can simply check if IF_FPM->MO_ADAPTATION_MANAGER is bound; simply
add the following line directly after the call of the parent’s GET_DATA method:
check CL_FPM_FACTORY=>GET_INSTANCE( )->MO_ADAPTATION_MANAGER is bound.

The next thing is to limit the additional FPM events needed for setting the adaptation context to those
situations where the relevant context has indeed changed. To do this, simply store the country value at every
event and in the next event check if it has changed. And only then, trigger the adaptation event. To do this,
add a private attribute MV_COUNTRY_CODE type SNWD_COUNTRY to your class, and add these lines
before you create the event:
check MV_COUNTRY_CODE ne <FS_COUNTRY_CODE>.

  MV_COUNTRY_CODE = <FS_COUNTRY_CODE>.

b. Setting the Adaptation Context Locally

To set the adaptation context for a single UIBB only, it’s possible to set only that UIBB locally. This can be
achieved by adding the UIBB Instance Key to the event and using an appropriate event.

Let’s start with using an appropriate event. There are two options:

Either use if_fpm_constants=>gc_event-ADAPT_CONTEXT_LOCAL as Event ID or any other event
(except if_fpm_constants=>gc_event-ADAPT_CONTEXT) and set its ADAPTS_CONTEXT attribute to
TRUE. If you then provide the UIBB’s configuration key as event parameter
if_fpm_constants=>gc_event_param-source_config_id, it’s automatically treated as a local
event.

In any case, for a local adaptation change we need the UIBB Instance Key. A simple solution would be to
hardcode this. However, this would then only work for the specified configuration and when the feeder is
reused in a different application this would not work. Therefore, we should try to determine the Instance Key
at runtime.

This information is passed to the feeder class in the IF_FPM_GUIBB~INITIALIZE method. So again, the
solution is obvious: Redefine this method, store the UIBB Instance information and then use it when firing the
event:

1. Add another private attribute to your feeder class: MS_INSTANCE_KEY type
FPM_S_CONFIG_KEY.

2. Redefine method IF_FPM_GUIBB~INITIALIZE and add the following code:

method IF_FPM_GUIBB~INITIALIZE.

CALL METHOD SUPER->IF_FPM_GUIBB~INITIALIZE
EXPORTING

      IT_PARAMETER = it_parameter
      IO_APP_PARAMETER = io_app_parameter
      IV_COMPONENT_NAME = iv_component_name
      IS_CONFIG_KEY = is_config_key
      IV_INSTANCE_ID = iv_instance_id.

move-CORRESPONDING IS_CONFIG_KEY to MS_INSTANCE_KEY.
  MS_INSTANCE_KEY-INSTANCE_ID = IV_INSTANCE_ID.

endmethod.

Trigger the local adaptation event. To do this, enhance the coding of your GET_DATA method by firing the
if_fpm_constants=>gc_event-ADAPT_CONTEXT_LOCAL event instead  of  the
if_fpm_constants=>gc_event-ADAPT_CONTEXT event, and add the Instance Key parameters to the
event. In the end, the whole GET_DATA method should look like this:

method IF_FPM_GUIBB_FORM~GET_DATA.

* Needed to convince feeder that data needs to be updated
if cs_data is initial.

      get_shared_data_instance( cl_epm_po_shared_data=>gc_class_name
)-

>add_all_change_notification( if_epm_po_header=>gc_bo_node_name ).
endif.

* call the GET_DATA method of the standard feeder class
call method super->if_fpm_guibb_form~get_data
exporting

        io_event = io_event
        it_selected_fields = it_selected_fields
        iv_raised_by_own_ui = iv_raised_by_own_ui
        iv_edit_mode = iv_edit_mode

importing
        et_messages = et_messages
        ev_data_changed = ev_data_changed
        ev_field_usage_changed = ev_field_usage_changed
        ev_action_usage_changed = ev_action_usage_changed

changing
        cs_data = cs_data
        ct_field_usage = ct_field_usage
        ct_action_usage = ct_action_usage.

* Only do the CBA processing if it's active
check cl_fpm_factory=>get_instance( )->mo_adaptation_manager is bound.

* Check event id to avoid infinite loop
check io_event->mv_event_id ne

                    if_fpm_constants=>gc_event-adapt_context_local.

data: lo_fpm type ref to if_fpm,
          lo_event type ref to cl_fpm_event.

field-symbols: <fs_country_code> type snwd_country.
* determine the country code from the data delivered by the standard feeder

assign component 'COUNTRY_CODE' of structure cs_data
to <fs_country_code>.

* Check that adaptation context needs to be adjusted.
check mv_country_code ne <fs_country_code>.

    mv_country_code = <fs_country_code>.

* create the object to set the adaptation context. The adaptation context
* is only set locally for the form.

create object lo_event
exporting

        iv_event_id = if_fpm_constants=>gc_event-adapt_context_local
        iv_adapts_context = abap_true.

* Set the adaptation context via event parameters
    lo_event->mo_event_data->set_value(

exporting iv_key = 'COUNTRY'
                                  iv_value = <fs_country_code> ).

* Add the information for which UIBB the adaptation context is set

    lo_event->mo_event_data->set_value(
exporting iv_key = if_fpm_constants=>gc_event_param-source_config_id

                 iv_value = ms_instance_key-config_id ).

    lo_event->mo_event_data->set_value(
exporting iv_key = if_fpm_constants=>gc_event_param-

source_config_type
                 iv_value = ms_instance_key-config_type ).

    lo_event->mo_event_data->set_value(
exporting iv_key = if_fpm_constants=>gc_event_param-

source_config_var
                 iv_value = ms_instance_key-config_var ).

    lo_event->mo_event_data->set_value(
exporting iv_key = if_fpm_constants=>gc_event_param-

source_instance_id
                 iv_value = ms_instance_key-instance_id ).

* finally, raise the event
    lo_fpm = cl_fpm_factory=>get_instance( ).
    lo_fpm->raise_event( lo_event ).

endmethod.

If you now run the application it will still behave as before, but now the adaptation context is only applied
locally to the Form UIBB.

Hiding UIBBs
In this section we want to prove that the adaptation context is only applied locally. This gives us the
opportunity to demonstrate another feature CBA offers, the dynamic hiding of UIBBs based on the
adaptation context.

Let’s create an adaptation for the List UIBB underneath the Form using the same adaptation context to see
that it isn’t applied. Navigate again via the application configuration to the List UIBB’s configuration, and
create an adaptation for ROLE = * COUNTRY = FR. In the Add Adaptation popup where you enter the
adaptation context, tick the check box Hide UIBB.

After you close the popup you are already done! By setting this checkbox you stated that this UIBB shall be
hidden if the adaptation context fits.

Now run the application again and choose a French supplier. It should be unchanged, the List UIBB should
still be there. Now let’s try it the other way round. Go to your GET_DATA method and change the code so
that the global adaptation event is thrown (simply delete the _LOCAL in the event id constant):
…
CREATE OBJECT LO_EVENT
EXPORTING

      IV_EVENT_ID = IF_FPM_CONSTANTS=>GC_EVENT-ADAPT_CONTEXT_LOCAL
      IV_ADAPTS_CONTEXT = abap_true.
…

When you execute the application now, the List UIBB should have been removed.

(Alternatively, you can set a breakpoint in the first statement after the CREATE OBJECT LO_EVENT
statement. Run the application again, select a French supplier and when the breakpoint is reached, change
the value of LO_EVENT->MV_EVENT_ID from FPM_ADAPT_CONTEXT_LOCAL to
FPM_ADAPT_CONTEXT. This way, we’re changing the event from local to global. Then let the program
continue).

Although this is probably a very useful feature, it was only introduced as a workaround for a restriction of
CBA (see following page).

Restriction:
Dynamic changes of the Adaptation Context at runtime will only affect adaptations of UIBB configurations.
The floorplan configurations are loaded at startup of the application and there is no way to replace them
later. Therefore, it’s not possible to change the UIBB assembly of a page at runtime via CBA. The only option
is the hiding of UIBBs.

So the natural approach to remove the List UIBB in the French adaptation context would be to create an
adaptation for the floorplan component for the adaptation context ROLE = * COUNTRY = FR (as in the first
part of this tutorial), and remove the List UIBB in this adaptation. But this will not work; the List UIBB will
remain on the screen!

